

Physics 364, Fall 2014, Lab #24 **Name:** _____

(Field Programmable Gate Arrays: 1)

Monday, November 24 (section 401); Tuesday, November 25 (section 402)

Course materials and schedule are at <http://positron.hep.upenn.edu/p364>

This file: <http://positron.hep.upenn.edu/wja/p364/2014/files/lab24.pdf>

Today's lab will introduce you to working with the BASYS2 circuit board, which is made by Digilent, Inc. The key component on the BASYS2 board is a small Field Programmable Gate Array, made by Xilinx, Inc. FPGAs can implement arbitrary combinations of logic gates (AND, OR, XOR, NAND, arithmetic, etc.), as well as flip-flops, memories, and more. The FPGA itself is the 1 cm × 1 cm chip at the center of the BASYS2 board. This little chip is equivalent to about 100,000 individual NAND gates! The board is both powered and programmed from the USB port of a computer (Windows or Linux, though I managed to get it to work at home on my Mac by using VirtualBox to run a Linux virtual machine). If you're curious for more details, the formal documentation for the BASYS2 board can be found at <http://www.digilentinc.com/Products/Detail.cfm?Prod=BASYS2> . The BASYS2 manual is at http://www.digilentinc.com/Data/Products/BASYS2/Basys2_rm.pdf .

There is very little for you to write up for today's lab, but there are a few questions here and there, which you should look out for. Also, before you leave class, you need to show us how far you got through the material. There are six parts: try to budget your time so that you get through all of them!

The FPGA is connected to 8 ON/OFF switches (lower edge of board, left side): the corresponding FPGA inputs are HIGH when the switches are UP and LOW when the switches are DOWN. Just above the 8 switches are 8 LEDs: the LEDs are ON when the corresponding FPGA outputs are HIGH and OFF when the outputs are LOW. To the right of the switches are 4 buttons: each one corresponds to an FPGA input that is normally LOW but goes HIGH while your finger is on the button. Above the buttons is a set of four 7-segment LED displays, which can be used to display numbers and a few letters, most commonly 0-9 and A-F for hexadecimal display of 16-bit numbers; these 32 LEDs (including the 4 decimal points) are somewhat confusingly controlled by 12 FPGA output pins. An on-board crystal oscillator provides a 25 MHz reference clock to the FPGA, which we will divide down and use at much lower speed. The board can connect to a PS/2 mouse or keyboard and to a VGA monitor, so in principle you could make a little video game, if you were really ambitious. On the top edge of the board are four sets of four user-defined input/output pins, which we will interface to our own breadboards in class next week.


The two main things that we have been aiming for this course to leave you with are (a) conceptual insight into how things work (a.k.a. demystification), and (b) technical skills that may help you to work in a research group. The reason for including the FPGA segment of the course is (a) to gain some conceptual understanding for how, in principle, a computer

can be build up out of logic gates (which you already know can be built up out of transistors), and (b) because FPGAs are a big part of the electronics design that we do for experiments in particle physics, medical physics, astronomy, and presumably other fields in which scientists build their own instruments.

Part 1

(a) Connect your board to the lab computer's USB port and turn it on by sliding the switch on the lower-left corner of the board to the UP position. If the blue "MODE" jumper on the top-right corner of the board is on the "ROM" (right-hand) position (but probably it is not, if your board has been used before), then you will see a 1111, 2222, 3333, ..., FFFF pattern count on the 7-segment LED display. If this is what you see, then power the board OFF, then move the blue jumper to the "PC" (left-hand) position, and power the board back ON. You should see a bright red LED illuminate near the power switch, but nothing else. Since nearly all of our BASYS2 boards were used in the Fall 2012 course, you probably will not need to do anything described in this paragraph.

Now start up the Digilent ADEPT software on the lab PC. The software should detect the BASYS2 board. From your web browser, download the file <http://positron.hep.upenn.edu/wja/p364/2014/files/ledcount.bit> to your PC (you'll need to pick a folder you're allowed to save to). Now in the ADEPT software's **config** tab, choose this `ledcount.bit` file as the program to load into the FPGA (XC3S100E). Click the **program** button! If all goes well, you should see the eight green LEDs counting in binary, updating once per second. If that doesn't work, ask for help! This is just a check that you can load a program onto the board. The ADEPT software (from Digilent, who make the board) is a tool for loading a `*.bit` file onto the ADEPT board. The `*.bit` file itself is created by the ISE software (from Xilinx, who make the FPGA chip).



(b) Now use your web browser to download

<http://positron.hep.upenn.edu/wja/p364/2014/files/lab24.zip>

which is an archived version of a Xilinx ISE project. (Again, you need to save the file in a location you are allowed to write to on the PC.) Now from Windows, right-click the file and extract its contents to a folder. Now go into that folder and double-click on `lab24.xise`, which should open up the Xilinx ISE software. In the ISE software's **processes** panel, right-click on **generate programming file** and choose **run**. After a moment, you should see the message Process "Generate Programming File" completed successfully. The result of all this is a file called `lab24_part1.bit`, which you can now load into the BASYS2 board using the ADEPT software. There is also a pre-compiled copy of `lab24_part1.bit` at http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part1_precompiled.bit

Here is the Verilog source code for the program that you just loaded:

```
1 //
2 // lab24_part1.v
3 // Verilog source code for Part 1 of Lab 24
4 // PHYS 364, UPenn, fall 2014
5 //
6
7 // This line forces you to declare the names of all wires explicitly;
8 // otherwise, if you use a name that the compiler is not aware of, it
9 // will quietly assume that it is a new wire name. In Verilog, a "wire"
10 // connects the output of one part of your circuit to the input of
11 // another part of your circuit, just like a physical wire.
12 `default_nettype none
13
14 // In Verilog, a "module" is like a schematic diagram that describes how
15 // a component of your circuit behaves. This "top-level" module represents
16 // the schematic diagram for the entire FPGA. Its inputs represent the
17 // signals coming in to the FPGA, and its outputs represent the signals
18 // going out of the FPGA. The signals with square brackets [] by their
19 // names are multi-bit signals: for example, there are 4 buttons and
20 // 8 switches among the FPGA's inputs, and there are 8 LEDs among the
21 // FPGA's outputs. The FPGA itself is the 1cm x 1cm chip sitting at the
22 // center of the BASYS2 printed circuit board. An FPGA is capable of
23 // implementing arbitrary digital logic functions, so it is a handy way
24 // for you to see what can be done with a very large collection of AND,
25 // OR, XOR gates, flip-flops, etc.
26 module lab24_part1
27 (
28     input      mclk, // 25 MHz clock built into the BASYS2 board
29     output [6:0] seg, // red 7-segment display to draw numbers & letters
30     output     dp,   // decimal point for 7-seg (0=ON, 1=OFF)
31     output [3:0] an, // shared LED anode signal per 7-seg digit (0=ON)
32     output [7:0] led, // 8 green LEDs, just above sliding switches
33     input  [7:0] sw,  // 8 sliding switches (up=1, down=0)
```

```

34  input  [3:0] btn,    // 4 push buttons (normal=0, pushed=1)
35  input  [4:1] ja,    // 4 pins (JA, NW corner) can connect to breadboard
36  input  [4:1] jb,    // 4 pins (JB) can connect to breadboard
37  input  [4:1] jc,    // 4 pins (JC) can connect to breadboard
38  input  [4:1] jd     // 4 pins (JD, NE corner) can connect to breadboard
39  );
40  // We define modules below called 'and_gate', 'nand_gate', etc. For
41  // each of these modules, the first argument is the output pin, and
42  // the next two arguments are the input pins. When I write the line
43  // 'and_gate myand1(led[0], sw[0], sw[1]);' it is equivalent to drawing
44  // an AND gate on my schematic diagram, writing the name 'myand1' to
45  // label this AND gate (so that it has a unique identity to distinguish
46  // two copies of the AND gate, e.g. 'myand1' and 'myand2'), and then
47  // connecting the output of the AND gate to LED #0 and the two inputs
48  // of the AND gate to switches #0 and #1.
49  and_gate myand1 (led[0], sw[0], sw[1]);
50
51  // Just to show you why we give each instance of 'and_gate' a unique
52  // name, here is a second copy of the same circuit (i.e. a second instance
53  // of the same 'and_gate' module), which functions equivalently to the
54  // first instance, but its output is connected to LED #7 and its input
55  // is connected to switches #6 and #7.
56  and_gate myand2 (led[7], sw[6], sw[7]);
57
58  // LED #1 should display the NAND of switches #0 and #1. Note that
59  // the switches and LEDs are numbered from 0 to 7, where 0 is on the
60  // right-hand side and 7 is on the left-hand side.
61  nand_gate mynand1 (led[1], sw[0], sw[1]);
62
63  // LED #2 should display the OR of the two switches.
64  or_gate  myor1   (led[2], sw[0], sw[1]);
65
66  // LED #3 should display the NOR of the two switches.
67  nor_gate mynor1 (led[3], sw[0], sw[1]);
68
69  // LED #4 should display the XOR of the two switches.
70  xor_gate myxor1 (led[4], sw[0], sw[1]);
71
72  // LED #5 is the 'Q' output of a SR latch, which is the most basic
73  // kind of flip-flop we studied. Button #0 is the SET input, and
74  // button #1 is the RESET input.
75  sr_latch mysr1  (led[5], btn[0], btn[1]);
76  assign led[6] = btn[3];
77
78  // These inputs are currently unused, but I connect them to some
79  // unnecessary logic so that the Xilinx compiler doesn't complain
80  // about unused inputs.

```

```

81     wire dummy = ({ja,jb,jc,jd,mclk}==0);
82
83     // These outputs are currently unused, but I connect them anyway to
84     // keep the compiler from complaining.
85     assign seg[6] = sw[6];
86     assign seg[5] = sw[5];
87     assign seg[4] = sw[4];
88     assign seg[3] = sw[3];
89     assign seg[2] = sw[2];
90     assign seg[1] = sw[1];
91     assign seg[0] = sw[0];
92     assign dp = dummy;
93     assign an[3] = ~btn[3];
94     assign an[2] = ~btn[2];
95     assign an[1] = ~btn[1];
96     assign an[0] = ~btn[0];
97 endmodule
98
99
100 // This 'and_gate' module tells the compiler what I mean when I ask it
101 // to put an 'and_gate' down on the top-level schematic. I define a module
102 // with one output pin called 'o' and two input pins called 'a' and 'b',
103 // to correspond with a real two-input AND gate.
104 module and_gate (output o, input a, input b);
105     // Verilog uses a C-like operator syntax. The '&' operator means the
106     // 'AND' operation. The assign statement tells the compiler to take
107     // whatever is on the right-hand side of the '=' sign and to wire it up
108     // to whatever is on the left-hand side. In this case, we 'AND' the two
109     // inputs and connect the result to the output.
110     assign o = a & b;
111 endmodule
112
113 module nand_gate (output o, input a, input b);
114     // We can also declare a 'wire' inside a module, which is like drawing
115     // a new wire on your schematic diagram. Writing it this way is like
116     // making a new wire called 'obbar' which is connected to the AND of
117     // inputs 'a' and 'b'. Then the wire 'obbar' is used as the input to
118     // a NOT gate (a.k.a. an inverter), whose output we connect to 'o'. In
119     // Verilog (as in C), the bit-wise NOT operator is a tilde (~).
120     wire obbar = a & b;
121     assign o = ~obbar;
122 endmodule
123
124 module or_gate (output o, input a, input b);
125     // In Verilog, as in C, the bit-wise OR operator is the vertical bar '|'.
126     assign o = a | b;
127 endmodule

```

```

128
129 module nor_gate (output o, input a, input b);
130     assign o = ~(a | b);
131 endmodule
132
133 module xor_gate (output o, input a, input b);
134     // In Verilog, as in C, you use a carat '^' for exclusive OR.
135     assign o = a ^ b;
136 endmodule
137
138 // This module implements the SR latch that we studied in Reading #11.
139 // You draw two NOR gates. The output of the first NOR gate is called 'q'.
140 // The output of the second NOR gate is called 'qbar' (i.e. the opposite
141 // of 'q'). The first NOR gate's inputs are connected to r (reset) and qbar.
142 // The second NOR gate's inputs are connected to s (set) and q. Normally,
143 // neither r nor s is asserted, and q keeps its previous state. If you
144 // assert r (but not s), then q goes to 0. If you assert s (but not r), then
145 // q goes to 1. An undefined state results if r and s are asserted together.
146 module sr_latch (output q, input s, input r);
147     // We define the wire 'qbar' here, but we don't connect it to anything,
148     // because using it as the output argument of the second nor_gate will
149     // connect it. You can see that the wire called 'qbar' connects the
150     // output of the second nor_gate to an input of the first nor_gate,
151     // just as you would do if you were drawing a schematic diagram of an
152     // SR latch.
153     wire qbar;
154     nor_gate mynor1 (q, r, qbar);
155     nor_gate mynor2 (qbar, s, q);
156 endmodule

```


As you learned in the reading, Verilog is a programming language that is used to describe digital logic systems. It is widely used for both FPGA programming and the design of integrated circuits. In this course, we'll just dabble in Verilog. If you're eager to learn more of the details when the course is over, an excellent introductory book is *FPGA Prototyping by Verilog Examples* by Pong P. Chu. I learned Verilog by reading Chu's book, so I can highly recommend it. Another book I have been meaning to read (but haven't yet), as it has good reviews on Amazon, is *Verilog by Example* by Blaine Readler.

<http://www.amazon.com/FPGA-Prototyping-Verilog-Examples-Spartan-3/dp/0470185325>
<http://www.amazon.com/Verilog-Example-Concise-Introduction-Design/dp/0983497303>

(c) Now let's explore what this Verilog program (called `lab24_part1.v`) is telling the FPGA to do.

- (i) The 8 green LEDs are numbered 0 to 7, from right to left. The 8 switches below the LEDs are also numbered 0 to 7, from right to left. Each switch is 1=UP, 0=DOWN. Move switches 0 and 1 back and forth to check that LED 0 shows the AND of switches 0 and 1. Check the behavior of the LED in response to the switches, and relate it to the Verilog source code. (You can also double-click `lab24_part1.v` in **sources** tab of ISE to open it up and edit it, if you like.)
- (ii) Check that LED 7 is the AND of switches 6 and 7.
- (iii) Now check that LED 1 is the NAND of switches 0 and 1.
- (iv) Check that LED 2 is the OR of switches 0 and 1.
- (v) Check that LED 3 is the NOR of switches 0 and 1, and again make sure you see how it is done.
- (vi) Check that LED 4 is the XOR of switches 0 and 1, and again look at the Verilog syntax.
- (vii) Now notice the SR latch (the first flip-flop-like device that we studied in today's reading), implemented using two NOR gates. LED 5 displays the state, **Q**, of the SR latch. Button 0 is the **set** input, and button 1 is the **reset** input. Check that the set and reset features work as you expect, and check that LED 5 remembers whether the most recent command was set vs. reset.
- (viii) Also notice that button 3 is connected directly to LED 6, so that you can see that the buttons are only on while you hold them down.

(d) Now let's ask the Xilinx ISE software to show us in schematic form its interpretation of this Verilog program. In the **processes** panel, click the (+) plus sign next to **Synthesize**, then double-click **View RTL Schematic**. (If you get a dialog box asking to choose between two display options, choose the second option ["Start with a schematic of the top-level block"] and uncheck the "show this dialog" option.) You will see a box indicating the inputs and outputs of the top-level FPGA program. Double-click the box to see what is inside. You will see what looks like a schematic diagram, with boxes for the various module instances like `myand1`. Double-click on each of these boxes to look inside. Here is what I see inside `myand1`, comfortably enough:



Look inside `mysr1` (both the schematic and the original Verilog) and make sure that you understand what became of the `wire qbar` that connects one NOR gate's output with an input of the other NOR gate.

Now hold down button 3 while you slide switches 0–6 up and down one at a time, and notice what happens to the segments of the red 7-segment display. Notice that each of the display's 7 segments is controlled by one switch.


Can you find a set of switch positions to spell out the numeral 5?

|

Now about the numeral 2?

|

Also notice what happens when you hold buttons 2 and 3 at the same time. The figure may help to explain what is going on. Notice that a single PNP transistor provides the power to each of the four 7-segment displays, while the 7 wires controlling the 7 segments of each LED are in fact shared between the four 7-segment displays. This is a trick that the manufacturer uses in order to reduce the number of required input/output pins. (I/O pins are often a precious resource in electronics.) If you want to use all four digits separately, you need to quickly (e.g. once per millisecond) alternate between the four PNP transistors, turning only one of them on at a time, while quickly wiggling the 7 segments for each LED digit. I'll do this for you next week.



Part 2

(To save time, you might just want to read through part 2 without actually doing anything. This is the least important part of today's lab.) Now replace the contents of your `lab24_part1.v` file with the contents of the following `lab24_part2.v`, which you can download from http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part2.v. The easiest way to do this is to edit the contents of your existing `lab24_part1.v` in ISE, then "select all," then delete. Now just click on the `lab24_part2.v` file in your web browser to display its contents. Click "control-A" to select all, then click "control-C" to copy. Now go back to ISE and click "control-V" to paste, then "control-S" to save. The result is that you've overwritten the old contents of `lab24_part1.v` with the online contents of `lab24_part2.v`. Overwriting the old contents of `lab24_part1.v` may be somewhat distasteful, but it is expedient.

```
1 //
2 // lab24_part2.v
3 // Verilog source code for Part 2 of Lab 24
4 // PHYS 364, UPenn, fall 2014
5 //
6
7 `default_nettype none
8
9 module lab24_part2
10 (
11     input        mclk, // 25 MHz clock built into the BASYS2 board
12     output [6:0] seg, // red 7-segment display to draw numbers & letters
13     output       dp, // decimal point for 7-seg (0=ON, 1=OFF)
14     output [3:0] an, // shared LED anode signal per 7-seg digit (0=ON)
15     output [7:0] led, // 8 green LEDs, just above sliding switches
16     input  [7:0] sw, // 8 sliding switches (up=1, down=0)
17     input  [3:0] btn, // 4 push buttons (normal=0, pushed=1)
18     input  [4:1] ja, // 4 pins (JA, NW corner) can connect to breadboard
19     input  [4:1] jb, // 4 pins (JB) can connect to breadboard
20     input  [4:1] jc, // 4 pins (JC) can connect to breadboard
21     input  [4:1] jd, // 4 pins (JD, NE corner) can connect to breadboard
22 );
23 assign led[0] = sw[0] & sw[1]; // LED0 is the AND of SW0 and SW1
24 assign led[7] = sw[6] & sw[7]; // LED7 is the AND of SW6 and SW7
25 assign led[1] = ~(sw[0] & sw[1]); // LED1 is the NAND of SW0 and SW1
26 assign led[2] = sw[0] | sw[1]; // LED2 is the OR of SW0 and SW1
27 assign led[3] = ~(sw[0] | sw[1]); // LED3 is the NOR of SW0 and SW1
28 assign led[4] = sw[0] ^ sw[1]; // LED4 is the XOR of SW0 and SW1
29 assign led[6] = btn[3]; // LED6 does whatever button 3 does
30
31 // LED5 is Q, BTN0 is SET, BTN1 is RESET, for SR latch
32 srlatch mysr1 (led[5], btn[0], btn[1]);
33
34 // These inputs are currently unused, but I connect them to some
```

```

35 // unnecessary logic so that the Xilinx compiler doesn't complain
36 // about unused inputs.
37 wire dummy = ({ja,jb,jc,jd,mclk}==0);
38
39 // These outputs are currently unused, but I connect them anyway to
40 // keep the compiler from complaining.
41 assign seg[6:0] = sw[6:0];
42 assign dp = dummy;
43 assign an[3:0] = ~btn[3:0];
44 endmodule
45
46
47 module srlatch (output q, input s, input r);
48 wire qbar = ~(q | s);
49 assign q = ~(qbar | r);
50 endmodule

```

Then double-click **generate programming file** to make a lab24_part2.bit output file. The behavior of this program is exactly the same as the behavior of lab24_part1.v . I mainly just want you to see two different ways of writing the same program — in one case, using many different modules to implement the various features, and in the other case, writing the equations directly in one module.

The other thing worth looking at here is the Xilinx compiler's schematic diagram representing this Verilog file. Again, in the **processes** panel, click the (+) plus sign next to **Synthesize**, then double-click **View RTL Schematic**. Then double-click the box representing the top-level FPGA program. Now the AND and OR gates are sitting at the top level. There are some inverters to convert the outputs into NAND and NOR. (**PLEASE NOTE:** it took me more than one try to get ISE to show me an up-to-date schematic diagram. I ended up doing **Project**→“Cleanup Project Files,” then quitting and restarting ISE, then re-running **Synthesize**, and then finally “View RTL Schematic.”)

Part 3

This time, let's create a new Xilinx project with the **New Project** wizard. First download these two source-code files:

http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part3.v

<http://positron.hep.upenn.edu/wja/p364/2014/files/basys2.ucf>


In the Xilinx ISE program, click **File** → **New Project**. Choose a project name and location, and set the top-level source type to “HDL.”

In the next dialog, set **Product Category** to “General Purpose,” set **Family** to “Spartan3E,” set **Device** to “XC3S100E,” set **Package** to “CP132,” and set **Speed** to “-4.” Click **next**.

In the **Project Summary** dialog, just click **finish**.

Now from the **Project** menu, choose **Add Source**. Add both `lab24_part3.v` and `basys2.ucf`, which you just downloaded a moment ago.

Now in the **processes** panel, right-click **Generate Programming File** and click **Process Properties**. Under the **Startup Options** category, change **FPGA Start-Up Clock** to “JTAG Clock,” and click **OK**.



Now in the **processes** panel, right-click **Generate Programming File** and click **Run**. The result (after a moment) will be a file `lab24_part3.bit` that you can load into the board using the ADEPT software. If you have trouble with the compiler, you can copy my `lab24_part3.bit` from

http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part3.bit

(a) The Verilog source code for `lab24_part3.v` is included below. Look through the code for the `add4bit` module and see if you can relate it to the 4-bit adder that we discussed in today's reading. Each bit of the sum is a 3-way XOR of two input bits and one carry bit, so that it is HIGH only if an odd number of these three bits are HIGH. Each carry bit is a 3-way OR, each term of which is an AND, such that the carry bit is HIGH only if two or more of these three bits are HIGH. Stare at this (and ask questions) until this makes sense.

```

1 //
2 // lab24_part3.v
3 // Verilog source code for Part 3 of Lab 24
4 // PHYS 364, UPenn, fall 2014
5 //
6
7 `default_nettype none
8
9 module lab24_part3
10 (
11     input      mclk, // 25 MHz clock built into the BASYS2 board
12     output [6:0] seg, // red 7-segment display to draw numbers & letters
13     output      dp,  // decimal point for 7-seg (0=ON, 1=OFF)
14     output [3:0] an, // shared LED anode signal per 7-seg digit (0=ON)
15     output [7:0] led, // 8 green LEDs, just above sliding switches
16     input  [7:0] sw,  // 8 sliding switches (up=1, down=0)
17     input  [3:0] btn, // 4 push buttons (normal=0, pushed=1)
18     input  [4:1] ja,  // 4 pins (JA, NW corner) can connect to breadboard
19     input  [4:1] jb,  // 4 pins (JB) can connect to breadboard
20     input  [4:1] jc,  // 4 pins (JC) can connect to breadboard
21     input  [4:1] jd   // 4 pins (JD, NE corner) can connect to breadboard
22 );
23
24 // These wires give understandable names to the inputs of
25 // the 4-bit adder, just to make the connections more clear.
26 wire a0 = sw[0]; // switches 0-3 will be a0, a1, a2, a3
27 wire a1 = sw[1];
28 wire a2 = sw[2];
29 wire a3 = sw[3];
30 wire b0 = sw[4]; // switches 4-7 will be b0, b1, b2, b3
31 wire b1 = sw[5];
32 wire b2 = sw[6];
33 wire b3 = sw[7];
34 wire cin = btn[3]; // push-button 3 will be carry-in
35 wire cout, s3, s2, s1, s0; // for the outputs of the 4-bit adder
36 add4bit myadder1 (cout, s3, s2, s1, s0,
37                 a3, a2, a1, a0,
38                 b3, b2, b1, b0, cin);
39 assign led[0] = s0; // LEDs 0-3 are the adder's SUM bits
40 assign led[1] = s1;
41 assign led[2] = s2;
42 assign led[3] = s3;
43 assign led[4] = cout; // LED4 is carry-out of the ader
44
45 // Assign dummy values to output pins that are currently unused
46 assign {led[7:5], seg[6:0], dp} = 0;
47 assign an[3:0] = 4'b1111;

```

```

48 endmodule
49
50
51 // This is the 4-bit adder circuit that we saw in
52 // the reading assignment. It's the second version,
53 // which includes carry-in and carry-out pins, so that
54 // in principle you can chain two of them together to
55 // make an 8-bit adder, etc.
56 module add4bit (output cout, output s3, output s2, output s1, output s0,
57               input a3, input a2, input a1, input a0,
58               input b3, input b2, input b1, input b0, input cin);
59   wire  c0  = (a0 & b0) | (a0 & cin) | (b0 & cin);
60   wire  c1  = (a1 & b1) | (a1 & c0) | (b1 & c0);
61   wire  c2  = (a2 & b2) | (a2 & c1) | (b2 & c1);
62   assign cout = (a3 & b3) | (a3 & c2) | (b3 & c2);
63   assign s0  = a0 ^ b0 ^ cin;
64   assign s1  = a1 ^ b1 ^ c0;
65   assign s2  = a2 ^ b2 ^ c1;
66   assign s3  = a3 ^ b3 ^ c2;
67 endmodule

```

(b) Now look at the lines in the main `lab24_part3` module where the `add4bit` module is *instantiated*, i.e. where the adder is plunked down onto the `lab24_part3` schematic. Notice that the adder inputs `a3 a2 a1 a0` come from switches 3–0 and that the adder inputs `b3 b2 b1 b0` come from switches 7–4. Then notice that the adder outputs `cout s3 s2 s1 s0` go to LEDs 4–0. Slide the switches up and down to spell out two 4-bit binary numbers A and B , and check that the adder displays the sum $A + B$ on the LEDs. You can also use push-button 3 to assert the `cin` input of the adder if you like. The reason that an adder has a `cin` input is so that you can e.g. chain together two 4-bit adders to make an 8-bit adder, and so on.


How did you check that the adder works as expected?

|

Part 4

Thus far (with just one exception), we've implemented purely **combinational** logic in the Xilinx FPGA, using the Verilog programming language. We've made a 4-bit adder add up the two 4-bit numbers specified by the positions of the 8 sliding switches, and we've displayed the adder's outputs on 5 green LEDs. The one piece of **sequential** logic that we implemented was a single SR latch, which could remember whether the *set* or *reset* button had been pressed more recently, as indicated by lighting (or not lighting) an LED to display the latch's **Q** state.

Let's continue now by programming the FPGA to implement the **4-bit counter** that you so tediously wired up in Lab 23 (though this version will include the carry-in bit as well):



This counter consists of a 4-bit adder and four **D-type flip-flops**. The counter's present value is represented by the states of the four flip-flop outputs: Q_3, Q_2, Q_1, Q_0 . On each clock cycle, the constant binary value 0001 is added to the counter's present value.

Here was the verilog code for the adder:

```
1 module add4bit (output cout, output s3, output s2, output s1, output s0,
2     input a3, input a2, input a1, input a0,
3     input b3, input b2, input b1, input b0, input cin);
4     wire c0 = (a0 & b0) | (a0 & cin) | (b0 & cin);
5     wire c1 = (a1 & b1) | (a1 & c0) | (b1 & c0);
```

```

6   wire   c2   = (a2 & b2) | (a2 & c1) | (b2 & c1);
7   assign cout = (a3 & b3) | (a3 & c2) | (b3 & c2);
8   assign s0   = a0 ^ b0 ^ cin;
9   assign s1   = a1 ^ b1 ^ c0;
10  assign s2   = a2 ^ b2 ^ c1;
11  assign s3   = a3 ^ b3 ^ c2;
12  endmodule

```

It turns out that the following recipe is how you tell Verilog to make a D-type flip-flop. For the moment, just treat it as a magic incantation. I will put some explanation of what it means into the reading. The basic idea is that whereas a **wire** merely connects two things together (like a wire on your breadboard), a **reg** can maintain an internal state. The statement

```
always @ (posedge clk) q_reg <= d;
```

tells Verilog that every time it detects a positive edge (i.e. a low-to-high transition) on the **clk** input, it should update **q_reg** to reflect the present state of the **d** input. That sounds just like the behavior of a DFF!

```

1  // This magic incantation is how you tell the Xilinx
2  // compiler to make a D-type flip-flop. Just accept
3  // it as an idiom, rather than trying at this stage
4  // to follow exactly what the syntax means. But the
5  // jist of it is that whenever a positive edge of the
6  // clk signal is seen, the contents of d are copied to
7  // the new contents of 'q_reg', which in turn is wired
8  // to the 'q' output of this circuit. We will use this
9  // module in parts 4 and 5.
10 module dflipflop (output q, input clk, input d);
11     reg q_reg;
12     always @ (posedge clk) q_reg <= d;
13     assign q = q_reg;
14 endmodule

```

Putting all of the pieces together looks like this:

```
1 //
2 // lab24_part4.v
3 // Verilog source code for Lab 24 part 4
4 // PHYS 364, UPenn, fall 2014
5 //
6
7 `default_nettype none
8
9 module lab24_part4
10 (
11     input    mclk, // 25 MHz clock built into the BASYS2 board
12     output [6:0] seg, // red 7-segment display to draw numbers & letters
13     output    dp, // decimal point for 7-seg (0=ON, 1=OFF)
14     output [3:0] an, // shared LED anode signal per 7-seg digit (0=ON)
15     output [7:0] led, // 8 green LEDs, just above sliding switches
16     input  [7:0] sw, // 8 sliding switches (up=1, down=0)
17     input  [3:0] btn, // 4 push buttons (normal=0, pushed=1)
18     input  [4:1] ja, // 4 pins (JA, NW corner) can connect to breadboard
19     input  [4:1] jb, // 4 pins (JB) can connect to breadboard
20     input  [4:1] jc, // 4 pins (JC) can connect to breadboard
21     input  [4:1] jd, // 4 pins (JD, NE corner) can connect to breadboard
22 );
23
24 // an[3] is Q, btn[0] is SET, btn[1] is RESET, for SR latch
25 wire srq;
26 srlatch mysr1 (srq, btn[0], btn[1]);
27 // make the left-most 7-segment LED light up if Q is true
28 assign an[3] = ~srq;
29
30 // We will use the 'Q' output of the SR latch as a clock
31 // for the counter's flip-flops, so that we can clock the
32 // counter by hand, one clock cycle at a time.
33 wire clock = srq;
34
35 // Define wires for the adder's S (sum) and COUT (carry) outputs
36 wire s3, s2, s1, s0, cout;
37 // The adder's CIN (carry in) will always be zero
38 wire cin = 0;
39 // The adder's B input will always be binary constant 0001
40 wire b3=0, b2=0, b1=0, b0=1;
41 // The adder's A inputs will be the flip flops' Q outputs
42 wire q3, q2, q1, q0;
43 add4bit myadder1 (cout,
44                 s3, s2, s1, s0,
45                 q3, q2, q1, q0,
46                 b3, b2, b1, b0, cin);
```

```

47 // Each flip-flop's D input will be one of the adder's
48 // S (sum) output bits
49 dflipflop mydff3 (q3, clock, s3);
50 dflipflop mydff2 (q2, clock, s2);
51 dflipflop mydff1 (q1, clock, s1);
52 dflipflop mydff0 (q0, clock, s0);
53
54 // Let the left four LEDs show the adder's S (sum) outputs
55 assign led[7] = s3;
56 assign led[6] = s2;
57 assign led[5] = s1;
58 assign led[4] = s0;
59
60 // Let the right four LEDs show the flip flops' Q outputs
61 assign led[3] = q3;
62 assign led[2] = q2;
63 assign led[1] = q1;
64 assign led[0] = q0;
65
66 // Light up all 7 segments (plus decimal) for any 7-segment digit
67 // that is activated. Digits 2,1,0 simply indicate whether buttons
68 // 2,1,0 are pressed. Digit 3 indicates the state of the SR latch
69 // (see comment above for an[3]).
70 assign seg[6:0] = 0;
71 assign dp = 0;
72 assign an[2:0] = ~btn[2:0];
73 endmodule
74
75
76 // four-bit adder
77 module add4bit (output cout, output s3, output s2, output s1, output s0,
78               input a3, input a2, input a1, input a0,
79               input b3, input b2, input b1, input b0, input cin);
80   wire c0 = (a0 & b0) | (a0 & cin) | (b0 & cin);
81   wire c1 = (a1 & b1) | (a1 & c0) | (b1 & c0);
82   wire c2 = (a2 & b2) | (a2 & c1) | (b2 & c1);
83   assign cout = (a3 & b3) | (a3 & c2) | (b3 & c2);
84   assign s0 = a0 ^ b0 ^ cin;
85   assign s1 = a1 ^ b1 ^ c0;
86   assign s2 = a2 ^ b2 ^ c1;
87   assign s3 = a3 ^ b3 ^ c2;
88 endmodule
89
90
91 // magic incantation to make a D-type flip-flop
92 module dflipflop (output q, input clk, input d);
93   reg q_reg;

```

```

94     always @ (posedge clk) q_reg <= d;
95     assign q = q_reg;
96 endmodule
97
98
99 // S-R latch (the simplest kind of flip-flop)
100 module srlatch (output q, input s, input r);
101     wire qbar;
102     nor mynorgate1 (qbar, q, s);
103     nor mynorgate2 (q, qbar, r);
104 endmodule

```

Download http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part4.v and use its contents to replace the present contents of your ISE project's `lab24_part1.v` (or whatever is the name of the Verilog source file in your open ISE project). If you don't already have an ISE project open from Part 3 (maybe you took a break after Part 3), you can find a `.zip` file containing a fresh ISE project at http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part4.zip whose contents you can just extract into a new folder, then double-click on `lab24_part4.xise` in that folder to open the ISE project. In case you have technical difficulties with the Xilinx compiler, I've put a pre-compiled version of `lab24_part4.bit` at http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part4.bit

Notice that we connected one of the adder's two inputs to the binary constant 0001 and the other input to the **Q** outputs of the four flip-flops. On each rising edge of the clock signal, the sum increments by 1.

Notice that the 4-bit output of the adder is displayed on LEDs 7,6,5,4, and notice that the 4 flip-flop **Q** outputs are displayed on LEDs 3,2,1,0.

For a clock signal, we are using the **Q** output of an SR latch, whose **reset** input is **button 1** and whose **set** input is **button 0**. You can monitor the present value of the latch's **Q** output by looking at the left-most digit of the 7-segment display. By alternatively pressing buttons 1 and 0, you can manually clock the counter.

Try clocking the counter by pushing (alternately) buttons 1 and 0.

Notice that the four **Q** bits from the DFFs are always one clock cycle behind the four **S** bits of the adder. In other words, the value displayed on the right 4 LEDs just after the clock edge is the value that had been displayed on the left 4 LEDs just before the clock edge.

Take the time to understand how the Verilog code is equivalent to the schematic diagram shown above.

Now **modify the program** so that the **B** input bits to the adder are the switch values `sw[3]`, `sw[2]`, `sw[1]`, `sw[0]` instead of the constant values 0, 0, 0, 1. Indicate below how you modified the program.

|

Now you will need to set the right four switches to DOWN, DOWN, DOWN, UP to get the counter to count up by ones. Give it a try! Can you make the counter count up by twos? By threes?


What settings should you use on the four switches to make your counter count DOWN by ones? Hint: do you remember the 4-bit two's complement representation for -1 ?

|

Part 5

Now we're going to connect 8 D-type flip-flops together into a **shift register**. This configuration is also known as a **pipeline**, because data flows from one flip-flop to the next on each clock cycle, as if it were moving through a pipe. The motion is also analogous to that of a factory assembly line.

The schematic diagram looks something like this, though I only drew 4 of the 8 flip-flops here:



Here is the Verilog code that is equivalent to the above schematic (but with 8 DFFs instead of just 4):

```
1 // The first flip-flop's D input will be sw[0] (the right-most
2 // switch). Each subsequent flip-flop's D input will be the
3 // previous flip-flop's Q output.
4 wire q7, q6, q5, q4, q3, q2, q1, q0;
5 dflipflop mydff0 (q0, clock, sw[0]);
6 dflipflop mydff1 (q1, clock, q0);
7 dflipflop mydff2 (q2, clock, q1);
8 dflipflop mydff3 (q3, clock, q2);
9 dflipflop mydff4 (q4, clock, q3);
10 dflipflop mydff5 (q5, clock, q4);
11 dflipflop mydff6 (q6, clock, q5);
12 dflipflop mydff7 (q7, clock, q6);
```

Notice that each flip-flop's **D** input is connected to the previous flip-flop's **Q** output. We also send all of the **Q** outputs to the 8 green LEDs:

```
1 // Display the 8 flip flops' Q values on the 8 green LEDs.
2 assign led[7] = q7;
3 assign led[6] = q6;
4 assign led[5] = q5;
5 assign led[4] = q4;
6 assign led[3] = q3;
7 assign led[2] = q2;
8 assign led[1] = q1;
9 assign led[0] = q0;
```

Here is the complete program, which you should download from http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part5.v You can paste the contents into the ISE editor in place of your previous contents of lab24_part4.v .

```
1 //
2 // lab24_part5.v
3 // Verilog source code for Lab 24 (part 5)
4 // PHYS 364, UPenn, fall 2014
5 //
6
7 `default_nettype none
8
9 module lab24_part5
10 (
11     input    mclk, // 25 MHz clock built into the BASYS2 board
12     output [6:0] seg, // red 7-segment display to draw numbers & letters
13     output    dp, // decimal point for 7-seg (0=ON, 1=OFF)
14     output [3:0] an, // shared LED anode signal per 7-seg digit (0=ON)
15     output [7:0] led, // 8 green LEDs, just above sliding switches
16     input  [7:0] sw, // 8 sliding switches (up=1, down=0)
17     input  [3:0] btn, // 4 push buttons (normal=0, pushed=1)
18     input  [4:1] ja, // 4 pins (JA, NW corner) can connect to breadboard
19     input  [4:1] jb, // 4 pins (JB) can connect to breadboard
20     input  [4:1] jc, // 4 pins (JC) can connect to breadboard
21     input  [4:1] jd // 4 pins (JD, NE corner) can connect to breadboard
22 );
23
24 // an[3] is Q, btn[0] is SET, btn[1] is RESET, for SR latch
25 wire srq;
26 srlatch mysr1 (srq, btn[0], btn[1]);
27 // make the left-most 7-segment LED light up if Q is true
28 assign an[3] = ~srq;
29
30 // We will use the 'Q' output of the SR latch as a clock
31 // for the counter's flip-flops, so that we can clock the
32 // counter by hand, one clock cycle at a time.
33 wire clock = srq;
34
35 // The first flip-flop's D input will be sw[0] (the right-most
36 // switch). Each subsequent flip-flop's D input will be the
37 // previous flip-flop's Q output.
38 wire q7, q6, q5, q4, q3, q2, q1, q0;
39 dfflipflop mydff0 (q0, clock, sw[0]);
40 dfflipflop mydff1 (q1, clock, q0);
41 dfflipflop mydff2 (q2, clock, q1);
42 dfflipflop mydff3 (q3, clock, q2);
43 dfflipflop mydff4 (q4, clock, q3);
44 dfflipflop mydff5 (q5, clock, q4);
```

```

45     dflipflop mydff6 (q6, clock, q5);
46     dflipflop mydff7 (q7, clock, q6);
47
48     // Display the 8 flip flops' Q values on the 8 green LEDs.
49     assign led[7] = q7;
50     assign led[6] = q6;
51     assign led[5] = q5;
52     assign led[4] = q4;
53     assign led[3] = q3;
54     assign led[2] = q2;
55     assign led[1] = q1;
56     assign led[0] = q0;
57
58     // Light up all 7 segments (plus decimal) for any 7-segment digit
59     // that is activated.  Digits 2,1,0 simply indicate whether buttons
60     // 2,1,0 are pressed.  Digit 3 indicates the state of the SR latch
61     // (see comment above for an[3]).
62     assign seg[6:0] = 0;
63     assign dp = 0;
64     assign an[2:0] = ~btn[2:0];
65 endmodule
66
67
68 // D-type flip-flop (treat this as a magic incantation for now)
69 module dflipflop (output q, input clk, input d);
70     reg q_reg;
71     always @ (posedge clk) q_reg <= d;
72     assign q = q_reg;
73 endmodule
74
75
76 // SR latch
77 module srlatch (output q, input s, input r);
78     wire qbar;
79     nor mynorgate1 (qbar, q, s);
80     nor mynorgate2 (q, qbar, r);
81 endmodule

```

Once you have the program compiled and loaded into your FPGA board, try clocking the flip-flops by alternately pressing buttons 1 and 0 (which alternately sets and resets the SR latch whose **Q** output supplies the clock signal). If you do this with SW0 in the DOWN position, you should see nothing very interesting happen.

Now set SW0 to the UP position. You should see the LEDs light up one-by-one.

Try setting SW0 back to the DOWN position. You should see darkness shift left clock-by-

clock in the same way as you saw light shifting left before.

Keep clocking in ZEROs until all of the LEDs are off. Now clock in a single ONE bit. In other words, flip SW0 to UP, then do a single clock cycle, then flip SW0 back to DOWN, then keep clocking. Do you see how the single ONE bit shifts from right to left? **Briefly explain what is happening.**



A shift register is sometimes used to delay data so that they can be kept around while a lengthy computation completes. The length of a shift register is analogous to the length of the *pipeline* on your home computer's microprocessor.

Part 6

Now here is a slightly different kind of D-type flip-flop. It has an additional input called `enable`. This is the type of flip-flop that I use most commonly in my own electronics-design work.

```
1 // This magic incantation is how you tell the Xilinx compiler to make
2 // a D-type flip-flop with the ENABLE feature.
3 module dffe (output q, input clk, input d, input enable);
4     reg q_reg;
5     always @ (posedge clk) if (enable) q_reg <= d;
6     assign q = q_reg;
7 endmodule
```

Now we're going to modify the main part of the Part 5 program to look like this instead:

```
1 // Now we set up 8 D-type flip-flops that have an additional
2 // input wire called ENABLE. If ENABLE is TRUE, then the
3 // flip-flop behaves exactly as an ordinary DFF. But if ENABLE
4 // is FALSE, then the DFF does nothing.
5
6 // Now we connect each flip flop's D input to the NOT of its own
7 // Q output. So if ENABLE is asserted, the flip-flop just toggles
8 // back and forth between 0 and 1. But if ENABLE is deasserted,
9 // the flip-flop does nothing.
10 wire q7, q6, q5, q4, q3, q2, q1, q0;
11 wire d7 = ~q7;
12 wire d6 = ~q6;
13 wire d5 = ~q5;
14 wire d4 = ~q4;
15 wire d3 = ~q3;
16 wire d2 = ~q2;
17 wire d1 = ~q1;
18 wire d0 = ~q0;
19
20 // Here is the tricky part: the first flip flop is always enabled.
21 // The second flip flop is only enabled if the first one is HIGH.
22 // The third flip flop is only enabled if the first two are HIGH.
23 // The fourth is only enabled if the first three are HIGH. Etc.
24 wire e0 = 1;
25 wire e1 = q0;
26 wire e2 = q0 & q1;
27 wire e3 = q0 & q1 & q2;
28 wire e4 = q0 & q1 & q2 & q3;
29 wire e5 = q0 & q1 & q2 & q3 & q4;
30 wire e6 = q0 & q1 & q2 & q3 & q4 & q5;
31 wire e7 = q0 & q1 & q2 & q3 & q4 & q5 & q6;
```

```
32     dffe mydff0 (q0, clock, d0, e0);
33     dffe mydff1 (q1, clock, d1, e1);
34     dffe mydff2 (q2, clock, d2, e2);
35     dffe mydff3 (q3, clock, d3, e3);
36     dffe mydff4 (q4, clock, d4, e4);
37     dffe mydff5 (q5, clock, d5, e5);
38     dffe mydff6 (q6, clock, d6, e6);
39     dffe mydff7 (q7, clock, d7, e7);
```

Can you guess what this program will do?! See if you can figure out what it is doing and why it works.



You can find the complete source code at http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part6.v Download this file and study it until it makes sense to you. Then paste its contents into your ISE editor in place of the contents of the Part 5 Verilog file that you have open. Then compile it and load the resulting `lab24_part6.bit` into your board. If you run into technical difficulties, you can find a pre-compiled `.bit` file at http://positron.hep.upenn.edu/wja/p364/2014/files/lab24_part6.bit